

Asymptotic Analysis & Sorting

The Fast and the Recursiveness

15-150 M21

Lecture 0609
09 June 2021

Today's slogan

Think big (and think long-term)

f is $O(1)$

- doubling the input size to f doesn't change the runtime

f is $O(n)$

- doubling the input size to f doubles the runtime

f is $O(\log n)$

- doubling the input size to f adds a constant to the runtime

f is $O(n^2)$

- doubling the input size to f quadruples the runtime

Check Your Understanding

What Big-O class is `pow` in?

- A $O(1)$
- B $O(\log n)$
- C $O(n)$
- D $O(n^2)$

Asymptotic Standpoint solves a few issues...

- Internal representations?

$$6+6 \implies ?^k 12 \quad (\text{for some constant } k)$$

- Hardware dependence: want algorithm analysis to be the same for both our computers, even if yours is twice as fast as mine. (constant coefficients are “consumed” by the Big-O)
- For small inputs, the costs of initiating the process and storing variables is much higher relative to the costs of actual computation (assuming asymptotically huge inputs, so not a problem)

Want: General method to
determine which Big-O class f is
in

- 0 How you're quantifying input size
- 1 Recurrence
- 2 Description of work tree
- 3 Measurements of work tree (height, and width at each level)
- 4 Summation
- 5 Big-O

Worked example: @ analysis

0609.0 (work.sml)

```
1 fun [] @ L2 = L2
2   | (x::xs) @ L2 = x::(xs@L2)
```

0 Measure of input size: length of the first list

1 Recurrence

Big Idea

Abstractly define $W : \mathbb{N} \rightarrow \mathbb{N}$ such that evaluating $f(x)$ for an input of size n takes approximately $W(n)$ steps, then classify the Big-O complexity of W

```
1 fun [] @ L2 = L2
2   | (x :: xs) @ L2 = x :: (xs @ L2)
```

0 Measure of input size: length of the first list

1 Recurrence:

$$W(0) = k_0$$


$$W(n) = k_1 + W(n - 1)$$


1 Recurrence:

$$W(0) = k_0$$

$$W(n) = k_1 + W(n - 1)$$

2 Work Tree





3 Measurements

Height: n

Work on the i -th level: k_1

4 Sum:

$$W(n) = k_0 + \sum_{i=0}^{n-1} k_1$$
$$\approx nk_1$$

5 Big O:

$W(n)$ is $O(n)$

```
1 (* op@ : t list * t list -> t list
2  * REQUIRES: true
3  * ENSURES: L1@L2 is the list consisting
4  *           of the elements of L1 (in order),
5  *           followed by the elements of L2
6  *           (in order)
7  * WORK: O(n), where n = |L1|
8  *)
9 fun [] @ L2 = L2
10  | (x::xs) @ L2 = x::(xs@L2)
```


Worked example: pow analysis

0609.1 (work.sml)

```
1 (* 0(1) *)  
2 fun square (x:int):int = x * x  
3  
4 (* 0(1) *)  
5 fun even (n : int):bool = (n mod 2)=0
```

0609.2 (work.sml)

```
1 fun pow 0 = 1
2   | pow n =
3     case (even n) of
4       true  => square(pow(n div 2))
5     | false => 2*square(pow(n div 2))
```

0609.2 (work.sml)

```
1 (* pow : int -> int
2  * REQUIRES: n >= 0
3  * ENSURES: pow(n) == exp(n)
4  * WORK: O(log n)
5  *)
6 fun pow 0 = 1
7   | pow n =
8     case (even n) of
9       true   => square(pow(n div 2))
10      | false => 2*square(pow(n div 2))
```

Check Your Understanding

Determine the Big-O complexity of `exp`

Worked example: rev analysis

0609.3 (work.sml)

```
1 (* rev : t list -> t list
2  * REQUIRES: true
3  * ENSURES: (rev L) == L', where L' is the list
4  *           containing the same elements as L,
5  *           but in the opposite order
6  * WORK:  $O(n^2)$ , where  $n = |L|$ 
7  *)
8 fun rev [] = []
9   | rev (x::xs) = (rev xs)@[x]
```

Check Your Understanding

Determine the Big-O complexity of `treev`

5-minute break

SML has a built-in type called `order`. It has three constructors/values:

`LESS` `EQUAL` `GREATER`

```
Int.compare : int * int -> order
```

```
String.compare : string * string -> order
```

```
fun quadrantV1 (m:int ,n:int):string =  
  if m=0 orelse n=0  
  then "boundary"  
  else if m>0  
        then if n>0  
              then "I"  
              else "IV"  
        else if n<0  
              then "II"  
              else "III"
```

```
fun quadrant (m:int, n:int):string =  
  case (Int.compare(m,0), Int.compare(n,0)) of  
    (EQUAL, _) => "boundary"  
  | (_, EQUAL) => "boundary"  
  | (GREATER, GREATER) => "I"  
  | (LESS, GREATER) => "II"  
  | (LESS, LESS) => "III"  
  | (GREATER, LESS) => "IV"
```

Sorting is a classic algorithmic problem in computer science: finding the fastest way to put all the elements of a list in order.

A value $[x_1, \dots, x_n] : \text{int list}$ is **sorted** if for each $i = 1, \dots, n - 1$, $\text{Int.compare}(x_i, x_{(i+1)}) \not\approx \text{GREATER}$.

Or, recursively: a value $v : \text{int list}$ is **sorted** if either $v = []$ or $v = [x]$ for some x , or $v = x :: x' :: xs$ where $\text{Int.compare}(x, x') \not\approx \text{GREATER}$ and $x' :: xs$ is sorted.

0609.4 (sorting.sml)

```
1 fun isSorted ([]:int list):bool = true
2   | isSorted [x] = true
3   | isSorted (x::x'::xs) =
4     (x<=x') andalso isSorted(x'::xs)
```

```
sort : int list -> int list
```

REQUIRES: true

ENSURES: sort (L) evaluates to a sorted permutation of L

A “permutation” of L is just a list that contains the same elements the same number of times as L, just in a possibly different order. So [1, 1, 2, 3] is a permutation of [3, 1, 2, 1] but not of [3, 2, 1].

There are many sorting algorithms: insertion sort, quick sort, merge sort, bubble sort, . . .

We'll be focusing on *merge sort*, which consists of the following three steps:

- 1 Split the input list in half
- 2 Sort each half
- 3 *merge* the sorted halves together to obtain a sorted whole


```
split : int list -> int list * int list
```

REQUIRES: true

ENSURES: `split L` evaluates to (A, B) where A and B differ in length by at most one, and $A@B$ is a permutation of L

```
merge : int list * int list -> int list
```

REQUIRES: A and B are sorted

ENSURES: `merge (A, B)` evaluates to a sorted permutation of $A@B$

```
msort : int list -> int list
```

REQUIRES: true

ENSURES: `msort (L)` evaluates to a sorted permutation of L

```
split : int list -> int list * int list
```

REQUIRES: true

ENSURES: `split L` evaluates to (A, B) where A and B differ in length by at most one, and $A@B$ is a permutation of L

0609.5 (sorting.sml)

```
1 fun split ([]):int list * int list = ([], [])
2   | split ([x] : int list) = ([x], [])
3   | split (x::x'::xs) =
4     let
5       val (A,B) = split xs
6     in
7       (x::A, x'::B)
8     end
```

```
merge : int list * int list -> int list
```

REQUIRES: A and B are sorted

ENSURES: merge (A , B) evaluates to a sorted permutation of A@B

0609.6 (sorting.sml)

```
1 fun merge (L1:int list , []:int list) = L1
2   | merge ([] , L2) = L2
3   | merge (x::xs , y::ys) =
4     (case Int.compare(x , y) of
5       GREATER => y::merge(x::xs , ys)
6       | _ => x::merge(xs , y::ys))
```

```
m_sort : int list -> int list
```

REQUIRES: true

ENSURES: m_sort (L) evaluates to a sorted permutation of L

0609.7 (sorting.sml)

```
1 fun m_sort ([] : int list) : int list = []  
2   | m_sort [x] = [x]  
3   | m_sort L =  
4     let  
5       val (A,B) = split L  
6     in  
7       merge(m_sort A,m_sort B)  
8     end
```

Analysis

0609.5 (sorting.sml)

```
1 fun split ([]):int list * int list = ([],[])
2   | split ([x] : int list) = ([x],[])
3   | split (x::x'::xs) =
4     let
5       val (A,B) = split xs
6     in
7       (x::A, x'::B)
8     end
```

0 Measure of size: length of input list

1-4 ...

5 $W_{\text{split}}(n)$ is $O(n)$

0609.6 (sorting.sml)

```
1 fun merge (L1:int list, []:int list) = L1
2   | merge ([], L2) = L2
3   | merge (x::xs, y::ys) =
4     (case Int.compare(x, y) of
5       GREATER => y::merge(x::xs, ys)
6       | _ => x::merge(xs, y::ys))
```

0 Measure of size: sum of lengths of input lists

1-4 ...

5 $W_{\text{merge}}(n)$ is $O(n)$

0609.7 (sorting.sml)

```
1 fun msort ([]:int list):int list = []
2   | msort [x] = [x]
3   | msort L =
4     let
5       val (A,B) = split L
6     in
7       merge(msort A,msort B)
8     end
```


0 Measure of size: length of input list

1

$$W_{\text{msort}}(0) = k_0$$

$$W_{\text{msort}}(1) = k_1$$

$$\begin{aligned} W_{\text{msort}}(n) &\leq k_2 + k_3n + W_{\text{msort}}(n/2) + W_{\text{msort}}(n/2) + k_4n \\ &\approx 2W_{\text{msort}}(n/2) + kn \end{aligned}$$

2 ...

3 Height: $\log n$, work on level i : $k(n - i)$

4 ...

5 $W_{\text{msort}}(n)$ is $O(n \log n)$

- We can make use of the formalism of recurrences and asymptotic complexity to precisely articulate the runtime of recursive functional functions/algorithms
- The Tree Method allows us to determine the asymptotic complexity of recursive functions.
- We can implement and analyze sorting algorithms using the tools we've developed so far

- Parallelism & Span
- Trees

Thank you!